

graph twoway pcspike — Paired-coordinate plot with spikes[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Reference](#)[Syntax](#)
[Also see](#)

Description

A paired-coordinate spike plot draws a spike (or line) for each observation in the dataset. The line starts at the coordinate (*y1var*, *x1var*) and ends at the coordinate (*y2var*, *x2var*).

Quick start

Paired-coordinate spike plot with spikes from (*y1*, *x1*) to (*y2*, *x2*)

```
twoway pcspike y1 x1 y2 x2
```

As above, with red lines

```
twoway pcspike y1 x1 y2 x2, lcolor(red)
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway pcp y1var x1var y2var x2var [if] [in] [, options]
```

| <i>options</i> | Description |
|----------------------------|--|
| <i>line_options</i> | change look of spike lines |
| vertical | orient plot naturally; the default |
| horizontal | orient plot transposing <i>y</i> and <i>x</i> values |
| <i>axis_choice_options</i> | associate plot with alternative axis |
| <i>twoway_options</i> | titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc. |

All explicit options are *rightmost*, except **vertical** and **horizontal**, which are *unique*; see [G-4] **concept: repeated options**.

Options

line_options specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] **line_options**.

vertical and **horizontal** specify whether the *y* and *x* coordinates are to be swapped before plotting—**vertical** (the default) does not swap the coordinates, whereas **horizontal** does.

These options are rarely used when plotting only paired-coordinate data; they can, however, be used to good effect when combining paired-coordinate plots with range plots, such as **twoway rspike** or **twoway rbar**; see [G-2] **graph twoway rspike** and [G-2] **graph twoway rbar**.

axis_choice_options associate the plot with a particular *y* or *x* axis on the graph; see [G-3] **axis_choice_options**.

twoway_options are a set of common options supported by all **twoway** graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] **twoway_options**.

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

Basic use

Advanced use

Advanced use 2

Basic use


We have longitudinal data from 1968 and 1988 on the earnings and total experience of U.S. women by occupation.

```
. use http://www.stata-press.com/data/r14/nlswide1
(National Longitudinal Survey. Young Women 14-26 years of age in 1968)
. list occ wage68 ttl_exp68 wage88 ttl_exp88
```

| | occ | wage68 | ttl_e~68 | wage88 | ttl_e~88 |
|----|--------------------|----------|----------|----------|----------|
| 1. | Professionals | 6.121874 | .860618 | 10.94776 | 14.11177 |
| 2. | Managers | 5.426208 | 1.354167 | 11.53928 | 13.88886 |
| 3. | Sales | 4.836701 | .9896552 | 7.290306 | 12.62823 |
| 4. | Clerical/unskilled | 4.088309 | .640812 | 9.612672 | 11.08019 |
| 5. | Craftsmen | 4.721373 | 1.091346 | 7.839769 | 12.64364 |
| 6. | Operatives | 4.364782 | .7959284 | 5.893025 | 11.99362 |
| 7. | Transport | 1.987857 | .5247414 | 3.200494 | 8.710394 |
| 8. | Laborers | 3.724821 | .775966 | 5.264415 | 10.56182 |
| 9. | Other | 5.58524 | .8278245 | 8.628641 | 12.78389 |

We graph a spike showing the movement from 1968 values to 1988 values for each observation (each occupation):


```
. twoway pcspike wage68 ttl_exp68 wage88 ttl_exp88
```



Advanced use

`twoway pcspike` can be usefully combined with other `twoway` plottypes (see [G-2] **graph twoway**). Here we add markers and labeled markers along with titles and such to improve the graph:


```
. twoway pcspike wage68 ttl_exp68 wage88 ttl_exp88      ||
  scatter wage68 ttl_exp68, msym(0)                  ||
  scatter wage88 ttl_exp88, msym(0) pstyle(p4)
  mlabel(occ) xscale(range(17))
  title("Change in US Women's Experience and Earnings")
  subtitle("By Occupation -- 1968 to 1988")
  ytitle(Earnings) xtitle(Total experience)
  note("Source: National Longitudinal Survey of Young Women")
  legend(order(2 "1968" 3 "1988"))
```



Advanced use 2


Drawing the edges of network diagrams is often easier with `twoway pcspike` than with other plottypes.

```
. use http://www.stata-press.com/data/r14/network1
. twoway pcspike y_c x_c y_l x_l
```



As with our first example, this graph can be made prettier by combining `twoway pcspike` with other plottypes.

```
. use http://www.stata-press.com/data/r14/network1a
. twoway pcspike y_c x_c y_l x_l, pstyle(p3) ||
    pcspike y_c x_c y_r x_r, pstyle(p4) ||
    scatter y_l x_l, pstyle(p3) msize(vlarge) msym(0)
        mlabel(lab_l) mlabpos(9) ||
    scatter y_c x_c, pstyle(p5) msize(vlarge) msym(0) ||
    scatter y_r x_r, pstyle(p4) msize(vlarge) msym(0)
        mlabel(lab_r) mlabpos(3)
yscale(off) xscale(off) ylabel(, nogrid) legend(off)
plotregion(margin(30 15 3 3))
```



Reference

Cox, N. J. 2009. Speaking Stata: Paired, parallel, or profile plots for changes, correlations, and other comparisons. *Stata Journal* 9: 621–639.

Also see

[G-2] [graph twoway](#) — Twoway graphs

[G-2] [graph twoway line](#) — Twoway line plots

[G-2] [graph twoway pcarrow](#) — Paired-coordinate plot with arrows

[G-2] [graph twoway pccapsym](#) — Paired-coordinate plot with spikes and marker symbols

[G-2] [graph twoway pci](#) — Twoway paired-coordinate plot with immediate arguments

[G-2] [graph twoway pcscatter](#) — Paired-coordinate plot with markers

[G-2] [graph twoway rspike](#) — Range plot with spikes